Paper 259-31

How Do I Look it Up If I Cannot Spell It:

An Introduction to SAS® Dictionary Tables

Peter Eberhardt, Fernwood Consulting Group Inc, Toronto, ON Ilene Brill, University of Alabama at Birmingham, Birmingham, AL

ABSTRACT

SAS maintains a wealth of information about the active SAS session, including information on libraries, tables, files and system options; this information is contained in the Dictionary Tables. Understanding and using these tables will help you build interactive and dynamic applications. Unfortunately, Dictionary Tables are often considered an 'Advanced' topic to SAS programmers. This paper will help novice and intermediate SAS programmers get started with their mastery of the Dictionary tables.

Ever needed a list of the tables (datasets) in a library? How about the columns (variables) in a table? Need to make sure you reset any titles after you run a report? Got some pesky warning messages in your SAS log you would like to clean up? Sure, you can look them up in the table and column properties in the explorer window. Or you can run a Proc Contents and check the listing. And of course you can ignore the warnings and errors in the SAS log since they 'almost always appear'. Or you can go to the Dictionary Tables and have your programme find out what libraries are allocated or what columns are available. So, what are Dictionary Tables and where do we access them?

Before we proceed, lets come to some common ground with terminology. In this paper we will talk about tables; for SAS programmers a table is the same as a dataset. Where a dataset has observations a table has rows. Where a dataset has variables a table has columns. You ask "Why do we use this terminology?". And we try to answer "Because Relational Database Management Systems (RDBMS) use this terminology, and they have always had their own Dictionary Tables. The SAS Dictionary Tables are documented in Proc SQL, so we assume this is why SAS uses the SQL/RDBMS terminology."

WHAT ARE DICTIONARY TABLES?

What happens when you start a SAS session? Ever right clicked on a table in the SAS explorer and looked at its properties? Ever wonder how you can get some of this neat information in a programme? Well, most of the good stuff is available in SAS Dictionary Tables. SAS Dictionary Tables are *read only* tables created and maintained by SAS; they contain a wealth of information about the current SAS session. In SAS v8 there were eleven Dictionary Tables; these were augmented in SAS v9, and now there are currently twenty-two Dictionary Tables that cover virtually all aspects of the SAS session. These Dictionary Tables are:

SAS V8 DICTIONARY TABLES

Table	Description	
CATALOGS	Contains information about SAS Catalogs	
COLUMNS	Contains information about variables/columns	
EXTFILES	Contains information about external files	
INDEXES	Contains information about columns participating in indexes	
MACROS	Contains information about specific to macros	
MEMBERS	Contains information about all data types (tables, views and catalogs)	
OPTIONS	Current session options	
STYLES	ODS styles	
TABLES	Contains information about tables/datasets	
TITLES	Contains information about titles and footnotes	
VIEWS	Contains information about views	

NEW SAS V9 DICTIONARY TABLES

Table	Description	
CHECK_CONSTRAINTS	Contains information about Check constraints	
CONSTRAINT_COLUMN_USAGE	Contains information about Constraint column usage	
CONSTRAINT_TABLE_USAGE	Constraint table usage	
DICTIONARIES	DICTIONARY tables and their columns	
ENGINES	Available engines	
FORMATS	Available formats	
GOPTIONS	SAS/Graph options	
LIBNAMES	LIBNAME information	
REFERENTIAL_CONSTRAINTS	Referential constraints	
REMEMBER	Remembered information	
TABLE_CONSTRAINTS	Table constraints	

Hint: Select * from dictionary.dictionaries; - lists all the tables given in the two tables above.

As we can see, many of the these dictionary tables contain the metadata about our data, and as SAS has evolved and added more RDBMS type capacity to its data management strengths we can see this reflected in new tables being added to the dictionary. In addition, SAS has provided us with tables with metadata about session environment. Before we look at how we can use these tables, let us look at the structure and content of each of the tables.

So how can we see the structure of these tables; that is, what columns are in the table? Proc SQL has a DESCRIBE TABLE command that will display the SQL that was used to create the table; the table structure is displayed in the log window. For example Listing 1 shows the SAS code submitted, SAS log notes and the table structure of DICTIONARY.TABLES as they appear in the log window:

LISTING 1 – GETTING THE STRUCTURE OF A DICTIONARY TABLE

```
390 proc sql;
391 describe table dictionary.tables;
NOTE: SQL table DICTIONARY. TABLES was created like:
create table DICTIONARY.TABLES
libname char(8) label='Library Name',
memname char(32) label='Member Name',
memtype char(8) label='Member Type',
memlabel char(256) label='Dataset Label',
typemem char(8) label='Dataset Type',
crdate num format=DATETIME informat=DATETIME label='Date Created',
modate num format=DATETIME informat=DATETIME label='Date Modified',
nobs num label='Number of Observations',
obslen num label='Observation Length',
nvar num label='Number of Variables',
protect char(3) label='Type of Password Protection',
compress char(8) label='Compression Routine',
encrypt char(8) label='Encryption',
npage num label='Number of Pages',
pcompress num label='Percent Compression',
reuse char(3) label='Reuse Space',
bufsize num label='Bufsize',
delobs num label='Number of Deleted Observations',
indxtype char(9) label='Type of Indexes',
datarep char(32) label='Data Representation',
requector char(24) format=$HEX informat=$HEX label='Requirements Vector'
 );
392 quit;
```

HOW DO I ACCESS DICTIONARY TABLES?

Now that we know what the table and column names are, how do we access them? There is a 'library' called DICTIONARY, so we access the tables the same way we access any SAS table using SQL. For example, to access the MEMBERS table we would do something like:

PROC SQL;
SELECT *
FROM dictionary.members;
QUIT;

A VIEW OF THE DICTIONARY

Now not only did the clever folks at SAS make these tables available, but they also made them even more accessible by creating views into these tables. The actual dictionary tables are not accessible in a DATA or PROC steps (aside from PROC SQL). The dictionary tables are in library called DICTIONARY, a 9 letter libref, and as we know, SAS librefs are limited to 8 characters so the views are needed to get access to the dictionary tables in DATA and PROC steps. These views are in the SASHELP library. The Dictionary Tables are only accessible through PROC SQL whereas, the views are accessible from any SAS proc or data step as well as the SAS explorer window. In this paper we use the term Dictionary Table though the examples use the equivalent views. These are views that are direct selections from the Dictionary Tables. These are:

View	Source
SASHELP.VCATALG	select * from dictionary.catalogs;
SASHELP.VCOLUMN	select * from dictionary.columns;
SASHELP.VEXTFL	select * from dictionary.extfiles;
SASHELP.VINDEX	select * from dictionary.indexes;
SASHELP.VMACRO	select * from dictionary.macros;
SASHELP.VMEMBER	select * from dictionary.members;
SASHELP.VOPTION	select * from dictionary.options;
SASHELP.VSTYLE	select * from dictionary.styles;
SASHELP.VTABLE	select * from dictionary.tables;
SASHELP.VTITLE	select * from dictionary.titles;
SASHELP.VVIEW	select * from dictionary.views;

Besides these views, SAS also creates more specialized views specifically targeted to the data tables. These views are:

View	Contents	Source
SASHELP.VSACCES	SAS/ACCESS Views	select libname, memname
		from dictionary.members
		where memtype = 'ACCESS'
		order by libname, memname;
SASHELP.VSCATLG	SAS CATALOGS	select libname, memname
		from dictionary.members
		where memtype = 'CATALOG'
		order by libname, memname;
SASHELP.VSLIB	SAS Libraries	select distinct(libname), path
		from dictionary.members
		order by libname;
SASHELP.VSTABLE	SAS Data Tables	select libname, memname
		from dictionary.members
		where memtype = 'DATA'
		order by libname, memname;
SASHELP.VSTABVW	SAS Data Tables and View	select libname, memname, memtype
		from dictionary.members
		where memtype ='VIEW'
		or memtype ='DATA'
		order by libname, memname;

SASHELP.VSVIEW	SAS Views	select libname, memname
		from dictionary.members
		where memtype = 'VIEW'
		order by libname, memname;

Hint: SASHELP.VSVIEW source code lists all the SAS view names given in the two tables above.

DICTIONARY TABLES IN MORE DEPTH

In this section we will look at some of the dictionary tables, first looking at the structure of the table and then talk about some of the ways it can be used. Our purpose here is not to enumerate each of the columns of each of the tables, but to give a general overview of the tables. We will start with the tables that have the metadata about our data tables and views.

DICTIONARY.MEMBERS

The MEMBERS table contains information about all the library member types - tables, views, and catalogs.

```
create table DICTIONARY.MEMBERS
  (
    libname char(8) label='Library Name',
    memname char(32) label='Member Name',
    memtype char(8) label='Member Type',
    dbms_memtype char(32) label='DBMS Member Type',
    engine char(8) label='Engine Name',
    index char(32) label='Indexes',
    path char(1024) label='Path Name'
    );
```

This table is a general overview of SAS libref. It can be used to determine the contents of a library, or perhaps to determine the type of a specific member. For example you can use the engine column to determine which version of SAS was used to create the library member or perhaps verify the location of the file by looking at the path column. With SAS libraries, the dbms_memtype is blank since it is described in the memtype column. However, for external databases (e.g. SQL Server/ODBC database), the memtype column says "DATA", and the dbms_memtype column tells whether it is a database table (value of "TABLE") or a database view (value of "VIEW").

DICTIONARY.TABLES

The TABLES table contains more detailed information about the members SAS thinks are tables/datasets; remember, for some external data sources SAS considers DBMS views as tables.

```
create table DICTIONARY.TABLES
  libname char(8) label='Library Name',
  memname char(32) label='Member Name',
  memtype char(8) label='Member Type',
  dbms_memtype char(32) label='DBMS Member Type',
  memlabel char(256) label='Dataset Label',
  typemem char(8) label='Dataset Type',
  crdate num format=DATETIME informat=DATETIME label='Date Created',
  modate num format=DATETIME informat=DATETIME label='Date Modified',
  nobs num label='Number of Physical Observations',
  obslen num label='Observation Length',
  nvar num label='Number of Variables',
  protect char(3) label='Type of Password Protection',
  compress char(8) label='Compression Routine',
  encrypt char(8) label='Encryption',
  npage num label='Number of Pages',
  filesize num label='Size of File',
  pcompress num label='Percent Compression',
```

```
reuse char(3) label='Reuse Space',
bufsize num label='Bufsize',
delobs num label='Number of Deleted Observations',
nlobs num label='Number of Logical Observations',
maxvar num label='Longest variable name',
maxlabel num label='Longest label',
maxgen num label='Maximum number of generations',
gen num label='Generation number',
attr char(3) label='Dataset Attributes',
indxtype char(9) label='Type of Indexes',
datarep char(32) label='Data Representation',
sortname char(8) label='Name of Collating Sequence',
sorttype char(4) label='Sorting Type',
sortchar char(8) label='Charset Sorted By',
requector char(24) format=$HEX48 informat=$HEX48
  label='Requirements Vector',
datarepname char(170) label='Data Representation Name',
encoding char(256) label='Data Encoding',
audit char(3) label='Audit Trail Active?',
audit_before char(3) label='Audit Before Image?',
audit_admin char(3) label='Audit Admin Image?',
audit_error char(3) label='Audit Error Image?',
audit_data char(3) label='Audit Data Image?'
```

The TABLES table is commonly used to get some basic information about the table such as number of rows (nobs) and/or columns (nvar) in the table, or the table creation/modification date. With some external data sources (e.g. ODBC) the nobs column is set to zero since the SAS/Access driver cannot return the number of rows in a table. Also, as noted above some DBMS views are reported in the TABLES table, although the dbms_memtype can be used to determine whether we are looking at a DBMS table or view.

DICTIONARY.VIEWS

The VIEWS table contains a more limited set of metadata about the views available. Note that it is reporting on SAS views, not views in external DBMSs.

```
create table DICTIONARY.VIEWS
  (
   libname char(8) label='Library Name',
   memname char(32) label='Member Name',
   memtype char(8) label='Member Type',
   engine char(8) label='Engine Name'
  );
```

Besides letting you determine which views are available, by looking at the engine column you can determine if the view was created as an SQL view or a DATA step view.

DICTIONARY.COLUMNS

The COLUMNS table provides detailed metadata about the columns in all of the tables and views.

```
create table DICTIONARY.COLUMNS
  (
    libname char(8) label='Library Name',
    memname char(32) label='Member Name',
    memtype char(8) label='Member Type',
    name char(32) label='Column Name',
    type char(4) label='Column Type',
    length num label='Column Length',
    npos num label='Column Position',
```

```
varnum num label='Column Number in Table',
label char(256) label='Column Label',
format char(49) label='Column Format',
informat char(49) label='Column Informat',
idxusage char(9) label='Column Index Type',
sortedby num label='Order in Key Sequence',
xtype char(12) label='Extended Type',
notnull char(3) label='Not NULL?',
precision num label='Precision',
scale num label='Scale',
transcode char(3) label='Transcoded?'
);
```

This table is commonly used to determine if a column exists (see examples below). It can also be used to verify the column type and format.

When looking at the above table layouts we see some common 'key' columns - particularly libname and memname. By joining the above three tables on these key columns it is possible to provide a custom report with your data dictionary.

In addition to the dictionary tables that describe your data, there are tables, which describe your SAS session. Let us look at some of these.

DICTIONARY.OPTIONS

The OPTIONS table has an entry for each of the SAS options.

```
create table DICTIONARY.OPTIONS
  (
    optname char(32) label='Option Name',
    opttype char(8) label='Option type',
    setting char(1024) label='Option Setting',
    optdesc char(160) label='Option Description',
    level char(8) label='Option Location',
    group char(32) label='Option Group'
    );
```

DICTIONARY.TITLES

The TITLES table has an entry for each title and footnote line currently in effect. See the example below on how to use this table to save the current titles, and then restore them after running a report.

```
create table DICTIONARY.TITLES
  (
   type char(1) label='Title Location',
   number num label='Title Number',
   text char(256) label='Title Text'
  );
```

DICTIONARY.EXTFILES

The EXTFILES table has an entry for each external file registered (filerefs) in the session.

```
create table DICTIONARY.EXTFILES
  (
   fileref char(8) label='Fileref',
   xpath char(1024) label='Path Name',
   xengine char(8) label='Engine Name'
);
```

This table is useful when you want to document external data sources/output from a run. Be aware that SAS has a number of filerefs it uses that you do not see in the explorer window; all of these SAS generated filerefs begin with #LN, so you can easily filter them out.

Now that we have looked at a few of the dictionary tables, let us look at some examples of how they could be used.

LOOKING IT UP IN THE DICTIONARY

You have developed an outstanding report that everyone wants included into their SAS runs. The problem is your report sets new title text and some users want their original titles after your report runs. Well, wrap your report in a macro and add two simple data steps, one before and one after your report (Listing 2).

LISTING 2 - RESETTING TITLES

```
%macro myreport;
/* use the dictionary to save all of the old titles and footnotes */
/* force the new title for demonstration purposes */
TITLE1 "The Original SAS Title1";
TITLE2 "The Original SAS Title2";
/* the SAS view sashelp.vtitle has the current titles and footnotes */
DATA oldtitles;
    set sashelp.vtitle;
RUN;
/* add a new title and run the report */
TITLE1 "This is a GREAT REPORT";
TITLE2 "With TWO Title Lines";
PROC SORT DATA=sashelp.shoes OUT=shoes;
     BY region product;
RUN;
PROC PRINT noobs DATA=shoes;
BY region product;
ID region product;
SUMBY product;
VAR sales returns;
RUN;
/* restore the titles from the dataset */
DATA null;
     SET __oldtitles END=done;
     length title $12.;
/* a title can be 200 chars, allow 2 extra for the quotes */
     length newtext $202.;
/* convert the title number to a char string */
     anum = compress(put(number, 2.));
/* put a single quote at the beginning of the text */
     newtext = "'" || text;
/* find the end of the text and add another quote */
     l = length(newtext);
     substr(newtext, l+1,1) = "'";
/* for titles, type = 'T' */
     if type = "T"
     then
       do;
        /* first make the string TITLE1 etc */
        title = "TITLE" || anum;
        /* now add the text so we end up with TITLE1 'This is the TEXT' */
        titletext = title || newtext;
        /* and put it out to the symbol table */
        call symput(title, titletext);
```

```
end;
     /* repeat for footnotes */
     else if type = "F"
     then
       do;
        /* first make the string FOOTNOTE1 etc */
        title = "FOOTNOTE" | anum;
        /* now add the text so we end up with FOOTNOTE1 'This is the TEXT' */
titletext = title || newtext;
        /* and put it out to the symbol table */
        titletext = title || newtext;
        call symput(title, titletext);
     end;
/* create a counter variable */
     if done
     then
       do:
         call symput("titlevars", put(_n_, 2.));
run;
%* now, pump out the old titles and footnotes ;
%do i = 1 %to &titlevars;
&&title&i;
%symdel title&I;
%end;
%symdel titlevars;
%mend myreport;
%myreport
```

Listing 2 can be viewed as a rough template for saving and restoring most settings in the SAS session. First, open the appropriate view (using a WHERE clause if appropriate) to save the current values. Set and use the new values. Finally, in another data step create the macro variables which are used to reset the original values.

A common use of the Dictionary Tables/Views is to identify and/or enumerate tables and columns available in the session. Lets take a quick look at viewing some of these metadata.

FIND THAT COLUMN

The Dictionary View SASHELP.VCOLUMN has the list of all the columns in all of the tables and views in your current SAS session. We can use this table to create a list of columns that are in multiple tables. First, we could do it with a simple listing (note the WHERE clause that excludes the MAPS and SASHELP libraries):

LISTING 3 - SELECTING COLUMNS IN MULTIPLE TABLES (1)

```
proc sql;
select name, count(*) as occurrences
from sashelp.vcolumn
where libname not in ('MAPS', 'SASHELP')
group by name
having count(*) > 1
order by name
;
quit;
```

Knowing the columns with multiple occurrences is useful, but it would be more useful to know in which tables the columns belong. With SAS this is where it gets interesting since there are usually a number of ways to solve the problem. One way is to create tables with the column names and use this to get more data on the columns:

LISTING 4 - SELECTING COLUMNS IN MULTIPLE TABLES (2)

```
proc sql;
create table MultiColCnt as
select name, count(*) as occurrences
from sashelp.vcolumn
where libname not in ('MAPS', 'SASHELP')
group by name
having count(*) > 1
order by name
;
create table MultiColTables as
select v.name, v.libname, v.memname, v.type, v.label, v.format
from sashelp.vcolumn as v, MultiColCnt as c
where c.name = v.name
and v.libname not in ('MAPS', 'SASHELP')
order by v.name, v.libname, v.memname
;
quit;
```

Now, suppose you want to apply a consistent label and format to a specific column that may be found in multiple tables. Well, the Dictionary Tables can help you locate the tables with the column and check whether the label and format need to be changed. For those that need changing, you can apply your favorite proc to change them. The following example uses a data step to select the tables that need changing then uses SQL to change them.

LISTING 5 - CHANGING THE LABEL AND FORMAT OF A SELECTED COLUMN

```
%macro ChangeLabelFormat(colName,
                                /* column to change */
                       newLabel,
                                /* label to apply
newFormat) /* format to apply */
%local tblsToChange;
%local i;
%* first locate the tables with the column;
%let tblsToChange = 0;
data null;
set sashelp.vcolumn (where=(upcase(name) EQ upcase("&colName") AND (label NE
"%newLabel" OR format NE "%newFormat") )
keep=libname memname name label format) end=done;
        = compress('ChqLib'
                              | | put(_n_, 7.)); chgTable =
compress('ChgTable' || put(_n_, 7.));
call symput(chgLib,
                     trim(libname));
call symput(chgTable, trim(memname));
if done
Then
do;
call symput('tblsToChange', put(_n_, 7.));
end;
```

```
run;
%* if any cols need changing, use SQL to change them;
%if &tblsToChange NE 0
%then
%do;
%do i = 1 %to &tblsToChange;
%let tblToChange = %cmpres(&&chgLib&i...&&chgTable&i);
proc sql;
alter table &tblToChange
modify &colName
label="&newLabel"
format=&newFormat
        ;
quit;
%end;
%end;
%mend;
```

CONCLUSION

Dictionary tables are an essential part of every SAS developer's toolbox. In the past, Michael Davis at SUGI 26 presented a paper that had a good overview and review of other papers showing different uses of the Dictionary Tables. Also, Pete Lund at SUGI 26 had an excellent paper on the use of the Dictionary Tables to document a project. This paper has provided a brief introduction to Dictionary Tables that we hope has helped you to better understand these concepts and thus they become more easily accessible to you as a useful tool.

REFERENCES

Ravi, Prasad. "Renaming All Variables in a SAS Data Set Using the Information from PROC SQL's Dictionary Tables"

SUGI 28 Seattle SAS Users Group International Proceedings. March 30-April 2, 2003. http://www2.sas.com/proceedings/sugi28/118-28.pdf

Beakley, Steven & McCoy, Suzanne. "Dynamic SAS Programming Techniques, or How NOT to Create Job Security"

SAS Conference Proceedings SUGI 29. May 2004. http://www2.sas.com/proceedings/sugi29/078-29.pdf

Lund, Pete. "A Quick and Easy Data Dictionary Macro" *SAS Conference Proceedings: SUGI 27.* April 2002. http://www2.sas.com/proceedings/sugi27/p099-27.pdf

Davis, Michael. "You Could Look it Up: An Introduction to SASHELP Dictionary Views" *SAS Conference Proceeding: SUGI 26.* April 2001. http://www2.sas.com/proceedings/sugi26/p017-26.pdf

Dilorio, Frank & Abolafia, Jeff. "Dictionary Tables and Views: Essential Tools for Serious Applications" SAS Conference Proceedings: SUGI 29. May 2004. http://www2.sas.com/proceedings/sugi29/237-29.pdf

ABOUT THE AUTHORS

Peter Eberhardt is SAS Certified Professional V8, SAS Certified Professional V6, and SAS Certified Professional - Data Management V6. In addition his company, Fernwood Consulting Group Inc. is a SAS Alliance Partner. Peter is a regular speaker at SUGI, SESUG and NESUG. If you have any questions or comments you can contact Peter at:

Fernwood Consulting Group Inc., 288 Laird Dr., Toronto ON M4G 3X5 Canada

Voice: (416)429-5705 e-mail: peter@fernwood.ca

Ilene Brill, a Systems Analyst for the Department of Epidemiology, University of Alabama at Birmingham (UAB), has used SAS for data management and analysis of STD epidemiology prospective follow-up studies, occupational epidemiology retrospective follow-up and case-control studies. Prior work includes the Department of Epidemiology and Biostatistics, University of Tromsø, Norway; Channing Laboratory, Boston, MA; Dept. of Epidemiology, Massachusetts Eye and Ear Infirmary; Dept.of Health Care Org. and Policy, UAB. M.P.H. Biometry, UAB; B.A. Biology, University of Pennsylvania.

Ilene Brill
Department of Epidemiology
University of Alabama at Birmingham
Ryals School of Public Health, Room 533B
1665 University Boulevard
Birmingham, AL 35294-0022

Work Phone: (205) 934-7160 Fax: (205) 975-7058

Email: <u>ibrill@ms.soph.uab.edu</u>

SAS, and SAS Alliance Partner are registered trademarks of SAS Institute Inc. in the USA and other countries. Other brand and product names are registered trademarks or trademarks of their respective companies.